

TD 9 - Analyse : transformée de Fourier discrète

La transformée de Fourier d'un signal s entre les instants 0 et T est donnée par :

$$\hat{s}(f) = \int_0^T s(t) e^{-2i\pi ft} dt$$

Si on approche cette intégrale par la méthode des rectangles avec un pas Δt égal à $\frac{1}{f_e}$ où f_e est une fréquence d'échantillonnage, on obtient la somme :

$$\hat{S}(f) = \sum_{k=0}^{N-1} \Delta t \times s(t_k) e^{-2i\pi ft_k} \text{ où } N = Tf_e \text{ et } t_k = \frac{k}{f_e}.$$

En prenant pour f les valeurs $f_n = \frac{n}{T}$ pour $n = 0, \dots, N-1$, on a :

$$\hat{S}(f_n) = \frac{1}{f_e} \sum_{k=0}^{N-1} s(t_k) e^{-2i\pi \frac{nk}{N}}$$

Cette dernière expression correspond à un facteur près à la définition suivante de la TFD d'une suite de nombres :

Définition


Soit $(x_k)_{0 \leq k \leq N-1}$ une suite de N nombres réels. On appelle transformée de Fourier discrète de cette suite, la suite $(y_n)_{0 \leq n \leq N-1}$ définie par :

$$y_n = \sum_{k=0}^{N-1} x_k e^{-2i\pi \frac{nk}{N}}$$

On considère maintenant un signal « porte », $s(t)$ défini par :


$$s(t) = \begin{cases} 1 & \text{si } |x| < 1 \\ 0 & \text{si } |x| > 1 \end{cases}$$

de représentation graphique :



Le module de la transformée de Fourier de ce signal est : $|\hat{s}(f)| = \left| \frac{\sin(\pi f)}{(\pi f)} \right|$.

représentation graphique :



Travail demandé :

Construire la TFD du signal carré et afficher le spectre des fréquences.

1. Discrétiser en temps à la fréquence d'échantillonnage f_e et construire la suite (x_k) de la définition ci-dessus.
2. Construire la suite (y_n) à partir de la suite (x_k) . On peut imbriquer deux boucles for ... end, une pour calculer la somme du membre de droite, l'indice n étant fixé, et une deuxième pour faire varier n .

Script Scilab

```
// Transformée de Fourier discrète
// -----
// définition d'une fonction "porte" de largeur 1
function s=f(t)
    if t<0 then s=0
    else if t<1 then s=1
        else s=0
    end
end
endfunction
// -----
utilisateur=x_mdialog('intervalle [a;b] et fréquence d\'échantillonnage',['a';'b';'fe'],['0';'4';'100'])
a=evstr(utilisateur(1));
b=evstr(utilisateur(2));
fe=evstr(utilisateur(3));
// nombre d'échantillon N du signal
N=(b-a)*fe;
// vecteur X contenant les échantillons
X=zeros(1,N);
for k=1:N
    X(k)=f(a+(k-1)/fe);
end
// vecteur Y contenant la TFD de la suite des valeurs dans X
Y=zeros(1,N);
v=exp(-2*i*pi/N);
w=1;
for n=1:N
    wk=1;
    for k=1:N
        Y(n)=Y(n)+wk*X(k);
        wk=wk*w;
    end
    w=w*v
end
// -----
// module des complexes de la suite Y
Z=abs(Y);
// -----
// échantillon F des fréquences
F=zeros(1,N);
for k=1:N
    F(k)=(k-1)*fe/N;
end
// symétrisation du spectre
M=floor(N/2);
Fsym=(fe/N).*[M-N+1:M];
Zsym=[Z(1,M+2:N),Z(1,1:M+1)];
// graphique
clf(0);
figure(0);
plot2d(Fsym,Zsym);
```